

User's manual to use thermodynamic libraries in GAMS through extrinsic functions

Juan I. Manassaldi^a, Miguel C. Mussati^{a,b}, Nicolás J. Scenna^a, Sergio F. Mussati^{a,b}

^a CAIMI Centro de Aplicaciones Informáticas y Modelado en Ingeniería (UTN-FRRO), Zeballos 1341, S2000BQA, Rosario, Argentina

^b INGAR Instituto de Desarrollo y Diseño (CONICET-UTN), Avellaneda 3657, S3002GJC, Santa Fe, Argentina

jmanassaldi@frro.utn.edu.ar, mmussati@santafe-conicet.gov.ar,
nscenna@santafe-conicet.gov.ar, mussati@santafe-conicet.gov.ar

1. Introduction

The main objective of this manual is to provide to all the users the basic steps to use thermodynamic libraries in GAMS employing extrinsic functions.

The steps are presented using specific cases including examples. However, it should be mentioned that the developed libraries can be used for any chemical/industrial process, according to the necessity of the users.

Three general-purpose thermodynamic libraries and one for water and steam have been developed and will be presented:

- *RaoultLaw.dll*: Ideal solution (liquid phase) + Ideal gas (vapor phase)
- *NRTLideal.dll*: NRTL activity coefficient (liquid phase) + Ideal gas (vapor phase)
- *PengRobinson.dll*: Peng Robinson equation of state (both phases)
- *IAPWS.dll*: The International Association for the Properties of Water and Steam (both phases).

2. General-purpose thermodynamic libraries (*RaoultLaw.dll*, *NRTLideal.dll*, *PengRobinson.dll*)

Below, general characteristics of the developed libraries are briefly summarized:

- Contain a database of 430 pure compounds.
- In a txt file, the IDs of the desired compounds and their interaction parameters (if necessary) should be defined.
- The input arguments of the function vary with the number of compounds involved.
- They support up to 18 compounds. Temperature + Pressure + 18 compounds = 20 argument (maximum arguments of extrinsic function for GAMS).
- All functions have as input arguments the Temperature, Pressure and molar fraction of each component of the mixture. For example, for a binary mixture, the functions will have 4 input arguments.
- All extrinsic functions have an analytic implementation of their gradient vector and Hessian matrix.
- Extrinsic Functions implemented in each library:

- Liquid and vapor phase density.
- Liquid and vapor phase enthalpy.
- Liquid and vapor phase entropy.
- Fugacity of each component in each phase (vapor and liquid).
- The database for pure compounds is taken from:
ChemSep v7.15 pure component data - Copyright (c) Harry Kooijman and Ross Taylor (2016) - http://www.perlfoundation.org/artistic_license_2_0
- The libraries were developed in *Dev C ++* and using *tdm-gcc* as a compiler.
- More information about the library can be found in the compilation section of the *lst* file.

2.1. Basic steps required for configure and use the libraries

2.1.1. Assignment of compounds

Before including any of the libraries, the desired compounds must be assigned in a *txt* file. As shown in Table 1, each library has a defined file name.

Table 1. ID filename for each library

Library	ID file name
<i>RaoultLaw.dll</i>	<i>RaoultLawID.txt</i>
<i>PengRobinson.dll</i>	<i>PengRobinsonID.txt</i>
<i>NRTLideal.dll</i>	<i>NRTLidealID.txt</i>

Then, the compounds desired by the users are defined from their ID in the pure compounds database (ChemSep v7.15 pure component data - Copyright (c) Harry Kooijman and Ross Taylor). Appendix 1 presents a list including the available compounds with their corresponding IDs.

An illustrative example considering the Peng Robinson's equation of state and five compounds (propane, isobutane, n-butane, isopentane and n-pentane) is shown below:

```
$onecho > PengRobinsonID.txt
ID1 3
ID2 4
ID3 5
ID4 8
ID5 7
$offecho
```

As indicated, only one space should be used to separate the compound number and its database ID.

2.1.2. Interaction parameter definition

Depending on the thermodynamic package selected, it is necessary to define a group of interaction parameters. Again, each library has a file name assigned to each interaction group, as shown in Table 2.

Table 2. Interaction parameters file name for each library

Library	Interaction parameters file name	Units
<i>PengRobinson.dll</i>	<i>PengRobinsonaij.txt</i>	unitless
<i>NRTLideal.dll</i>	<i>NRTLidealaij.txt</i>	cal/mol
	<i>NRTLidealalphaij.txt</i>	unitless

Only binary interaction parameters that are not repeated should be defined. Interaction parameters definition for the previously defined mixture is shown below.

```
$onecho > PengRobinsonaij.txt
a12 -0.0078
a13 0.0033
a14 0.0111
a15 0.0267
a23 -0.0004
a24 0.0005043
a25 0.00067951
a34 0.00021669
a35 0.0174
a45 0.06
$offecho
```

As example, the definition of parameters corresponding to a mixture consisting of ethanol and water using the *NRTLideal.dll* library is shown below.

```
$onecho > NRTLidealID.txt
ID1 1921
ID2 1102
$offecho
$onecho > NRTLidealaij.txt
a12 -57.9601
a21 1241.7396
$offecho
$onecho > NRTLidealalphaij.txt
alpha12 0.2937
$offecho
```

As shown (for *NRTLideal.dll*), the values of the parameters a_{12} and a_{21} must be defined because they are different. But, α_{12} and α_{21} have the same values.

Therefore, α_{12} is only defined and then the library assigns internally the same values for α_{21} .

2.1.3. Including the developed libraries in GAMS

The following internal coding is used to include the libraries into GAMS:

```
$FuncLibIn <InternalLibName> <ExternalLibName>
```

Therefore, the *NRTLideal.dll* library is included as follows:

```
$FuncLibIn NRTLideal NRTLideal.dll
```

The *NRTLideal.dll* file must be placed in the subdirectory *gamsdir/projdir*, otherwise, the corresponding fullpath must be specified. The library must be included after the definition of the compounds and interaction parameters.

Once the library is included, the functions arguments are automatically assigned to specify temperature, pressure, and compositions. Thus, the total number of arguments required depends on the number of compounds. For instance, the arguments needed for a binary mixture are four (T , P , x_1 , x_2). In the compilation section in the *lst* file, it is possible to check if the compounds have been well identified. For example, the following information corresponds to *NRTLideal.dll* library execution (ethanol and water mixture).

```
FUNCLIBIN NRTLideal NRTLideal.dll
Function Library NRTLideal
NRTL + IG Property Package v0.9 by Ph.D. J.I. Manassaldi (jmanassaldi@frro.utn.edu.ar); Ph.D. N.J. Scenna; Ph.D. M.C. Mussati; Ph.D. S.F. Mussati (mussati@santafe-conicet.gov.ar)
GAMS Development Corporation

Mod. Function Description
Type

NLP rho_liq(temperature [k],pressure [bar],water,ethanol)liquid phase molar density [mol/m3]
NLP rho_vap(temperature [k],pressure [bar],water,ethanol)vapor phase molar density [mol/m3]
NLP h_liq(temperature [k],pressure [bar],water,ethanol)liquid phase molar enthalpy [J/mol]
NLP h_vap(temperature [k],pressure [bar],water,ethanol)vapor phase molar enthalpy [J/mol]
NLP s_liq(temperature [k],pressure [bar],water,ethanol)liquid phase molar entropy [J/(mol.K)]
NLP s_vap(temperature [k],pressure [bar],water,ethanol)vapor phase molar entropy [J/(mol.K)]
NLP f1_liq(temperature [k],pressure [bar],water,ethanol)liquid phase fugacity of component 1 [bar]
NLP f1_vap(temperature [k],pressure [bar],water,ethanol)vapor phase fugacity of component 1 [bar]
NLP f2_liq(temperature [k],pressure [bar],water,ethanol)liquid phase fugacity of component 2 [bar]
NLP f2_vap(temperature [k],pressure [bar],water,ethanol)vapor phase fugacity of component 2 [bar]
```

To avoid inconsistencies, it is important to observe the units of the input and output arguments of the functions.

2.1.4. Functions definition

After the library is included, the necessary functions must be defined. This task is also done using an internal coding of GAMS, which is indicated below:

```
function <InternalFuncName> /<InternalLibName>.<FuncName>/;
```

Thus, by applying the above internal code, the function for *liquid enthalpy* corresponding to the *NRTLideal.dll* library is defined as follows:

```
function hliq /NRTLideal.h_liq/;
```

In this example, for convenience the original extrinsic function *h_liq* was redefined as *hliq*.

2.1.5. Usage of libraries

Once the previous steps have been completed, the developed extrinsic functions are already available for use. They can be used to define parameters or include them in an equation.

In this section, an illustrative optimization example to show a detailed application of one of the developed libraries (*NRTLideal.dll*) is presented.

The objective of the optimization problem is to calculate the composition and temperature of a binary minimum-boiling homogeneous azeotrope. Precisely, a water-ethanol mixture is considered and the pressure is fixed at 1.0132 bar.

To do this, the following mathematical model has been used (*Azeotrope.gms*).

```
$onecho > NRTLidealID.txt
ID1 1921
ID2 1102
$offecho
$onecho > NRTLidealaij.txt
a12 1241.7396
a21 -57.9601
$offecho
$onecho > NRTLidealalphaij.txt
alpha12 0.2937
$offecho
$funclibin NRTLideal NRTLideal.dll
function f1l /NRTLideal.f1_liq /;
function f2l /NRTLideal.f2_liq /;
function f1v /NRTLideal.f1_vap /;
function f2v /NRTLideal.f2_vap /;

sets
i compounds /water,ethanol/
;
Parameter
P pressure [bar] /1.0132/
;
Variable
T temperature [K]
x(i) liquid molar fraction
y(i) vapor molar fraction
;
equation
eq1,eq2 phase equilibrium equations
```

```

eq3,eq4 sumatory of component molar fractions
;
eq1.. f1l(T,P,x('water'),x('ethanol')) =e= f1v(T,P,y('water'),y('ethanol'));
eq2.. f2l(T,P,x('water'),x('ethanol')) =e= f2v(T,P,y('water'),y('ethanol'));
eq3.. sum(i,y(i)) =e= 1;
eq4.. sum(i,x(i)) =e= 1;

y.lo(i)=0; y.up(i)=1;
x.lo(i)=0; x.up(i)=1;

y.l(i)=0.5;
x.l(i)=0.5;
T.l=350;

model azeotrope /all/;
solve azeotrope using nlp minimizing T;

```

The obtained results are compared (Table 3) with experimental data taken from Tochigi et al. (1985). Despite the assumptions made to derive the model, for instance, the dependence of the NRTL interaction parameters with the temperature is neglected, a good agreement between the predicted and experimental results is observed.

Table 3. Comparison of the model-based results and experimental value (1.0132 bar)

	Implemented GAMS model	Experimental data ^a
Temperature [K]	351.5302	351.34
Composition (ethanol molar fraction)	0.8681	0.894

^a Tochigi, K., Inoue, H., and Kojima, K. (1985). Determination of azeotropes in binary systems at reduced pressures. *Fluid Phase Equilibria* 22, 343–352.

3. Properties of Water and Steam library (IAPWS.dll)

The *IAPWS.dll* library has been developed based on the correlations presented in “Revised Release on the IAPWS Industrial Formulation 1997 for the Thermodynamic Properties of Water and Steam” (<http://www.iapws.org/relguide/IF97-Rev.pdf>)

Some characteristics of the developed library are detailed below:

- This formulation is recommended for industrial use (primarily the steam power industry) for the calculation of thermodynamic properties of ordinary water in its fluid phases, including vapor-liquid equilibrium.
- The range of validity is divided into several regions (with close tolerances for consistency at region boundaries), each of which is represented by a different fundamental equation.
- Only regions 1 (liquid), 2 (vapor) and 4 (saturated line) were implemented. They correspond to those of greater use.
- All the extrinsic functions have an analytic implementation of their gradient vector and Hessian matrix.
- Implemented functions

- Saturation temperature equation (Region 4).
- Specific enthalpy as a function of pressure and temperature (Region 1 and 2).
- Specific entropy as a function of pressure and temperature (Region 1 and 2).
- Specific volume as a function of pressure and temperature (Region 1 and 2).
- Specific enthalpy as a function of pressure and vapor fraction.
- Specific entropy as a function of pressure and vapor fraction.

3.1. Basic steps required for use the library

3.1.1. Inclusion of the IAPWS.dll library in GAMS

In this case, it is not necessary to define any parameter before including the library *IAPWS.dll*. It is included in the same way as any other library.

```
$funclibin IAPWS IAPWS.dll
```

In the *lst* file, the information of the compilation stage associated with the *IAPWS.dll* library reads as follows:

```
FUNCLIBIN IAPWS D:\Manassaldi\CAIMI\Equipos en C\libreria de
equipos\IAPWS new\IAPWS.dll
Function Library IAPWS
IAPWS Property Package v0.9 by Ph.D. J.I. Manassaldi
(jmanassaldi@frro.utn.edu.ar); Ph.D. N.J. Scenna; Ph.D. M.C. Mussati;
Ph.D. S.F. Mussati (mussati@santafe-conicet.gov.ar)
GAMS Development Corporation

Mod. Function Description
Type

NLP Tsat_p(p [bar]) Saturation Temperature [K]
NLP Hliq_pt(p [bar],t [k]) Liquid Enthalpy [kJ/kg]
NLP Hvap_pt(p [bar],t [k]) Vapor Enthalpy [kJ/kg]
NLP Htit_px(p [bar],x) Saturated Vapor Enthalpy [kJ/kg]
NLP sliq_pt(p [bar],t [k]) Liquid Entropy [kJ/(kg.K)]
NLP svap_pt(p [bar],t [k]) Vapor Entropy [kJ/(kg.K)]
NLP stit_px(p [bar],x) Saturated Vapor Entropy [kJ/(kg.K)]
NLP vliq_pt(p [bar],t [k]) Liquid Specific Volume [m3/kg]
NLP vvap_pt(p [bar],t [k]) Vapor Specific Volume [m3/kg]
```

Again, to avoid unit inconsistencies it is important to observe the units of the input and output arguments of the functions.

3.1.2. Illustrative example of the usage of IAPWS.dll library

An illustrative example to find the maximum value of the saturated steam enthalpy is presented below (*saturatedvapor.gms*).

```

$funclibin IAPWS IAPWS.dll
function Tsat_p /IAPWS.Tsat_p/
function Hvap_pt /IAPWS.Hvap_pt/
;
Variable
P pressure [bar]
T temperature [K]
h especific enthalpy [kJ(kg-1)]
;
equation
eq1 saturation condition
eq2 enthalpy definition
;
eq1.. T =e= Tsat_p(P);
eq2.. h =e= Hvap_pt(P,T);

P.lo=1;          P.up=60;
T.lo=Tsat_p(P.lo); T.up=Tsat_p(P.up);

model saturatedvapor /all/
solve saturatedvapor usning nlp maximizing h;

```

Figure 1 illustrates the optimal value of the pressure that imply the maximum value of the saturated steam enthalpy (red point) and the plot of the enthalpy vs. pressure (continuous line) obtained by varying parametrically the pressure (simulation mode). As expected, the comparison shows that the model result is in agreement to that obtained parametrically.




Figure 1. Saturated steam enthalpy.

Appendix 1. Supported compounds with the corresponding IDs.

ID	Name	ID	Name	ID	Name
1	Methane	505	O-xylene	1319	Isopropyl acetate
2	Ethane	506	M-xylene	1321	Vinyl acetate
3	Propane	507	P-xylene	1322	Methyl propionate
4	Isobutane	509	N-propylbenzene	1351	Methyl methacrylate
5	N-butane	510	Cumene	1357	N-pentyl acetate
7	N-pentane	511	O-ethyltoluene	1363	N-hexyl acetate
8	Isopentane	512	M-ethyltoluene	1366	Ethylene carbonate
9	Neopentane	513	P-ethyltoluene	1381	Dimethyl terephthalate
11	N-hexane	514	1,2,3-trimethylbenzene	1401	Dimethyl ether
12	2-methylpentane	515	1,2,4-trimethylbenzene	1402	Diethyl ether
13	3-methylpentane	516	Mesitylene	1403	Diisopropyl ether
14	2,2-dimethylbutane	518	N-butylbenzene	1404	Di-n-butyl ether
15	2,3-dimethylbutane	519	Isobutylbenzene	1405	Methyl tert-butyl ether
17	N-heptane	520	Sec-butylbenzene	1406	Di-sec-butyl ether
18	2-methylhexane	521	Tert-butylbenzene	1407	Methyl ethyl ether
19	3-methylhexane	522	O-cymene	1408	Methyl n-propyl ether
20	3-ethylpentane	523	M-cymene	1409	Isopropyl butyl ether
21	2,2-dimethylpentane	524	P-cymene	1410	Methyl isobutyl ether
22	2,3-dimethylpentane	525	O-diethylbenzene	1411	Methyl isopropyl ether
23	2,4-dimethylpentane	526	M-diethylbenzene	1421	1,4-dioxane
24	3,3-dimethylpentane	527	P-diethylbenzene	1427	Methyl tert-pentyl ether
25	2,2,3-trimethylbutane	530	1,2,3,4-tetramethylbenzene	1428	Tert-butyl ethyl ether
27	N-octane	531	1,2,3,5-tetramethylbenzene	1430	Ethyl tert-pentyl ether
28	2-methylheptane	532	1,2,4,5-tetramethylbenzene	1431	Methylal
29	3-methylheptane	544	P-diisopropylbenzene	1441	Ethylene oxide
30	4-methylheptane	558	Biphenyl	1442	1,2-propylene oxide
31	3-ethylhexane	576	2-ethyl-m-xylene	1447	Butyl vinyl ether
32	2,2-dimethylhexane	577	2-ethyl-p-xylene	1461	Anisole
33	2,3-dimethylhexane	578	4-ethyl-m-xylene	1472	Cumene hydroperoxide
34	2,4-dimethylhexane	579	4-ethyl-o-xylene	1479	Tetrahydrofuran
35	2,5-dimethylhexane	586	1-methyl-3-n-propylbenzene	1501	Carbon tetrachloride
36	3,3-dimethylhexane	587	1-methyl-4-n-propylbenzene	1502	Methyl chloride
37	3,4-dimethylhexane	601	Styrene	1503	Ethyl chloride
38	2-methyl-3-ethylpentane	701	Naphthalene	1504	Vinyl chloride
39	3-methyl-3-ethylpentane	702	1-methylnaphthalene	1521	Chloroform
40	2,2,3-trimethylpentane	703	2-methylnaphthalene	1522	1,1-dichloroethane
41	2,2,4-trimethylpentane	710	1-phenylnaphthalene	1523	1,2-dichloroethane
42	2,3,3-trimethylpentane	717	Fluoranthene	1524	1,1,2-trichloroethane
43	2,3,4-trimethylpentane	723	1-methylindene	1541	Trichloroethylene
44	2,2,3,3-tetramethylbutane	724	2-methylindene	1571	Monochlorobenzene
46	N-nonane	738	Fluorene	1572	O-dichlorobenzene

47	2,2,5-trimethylhexane	803	Indene	1573	M-dichlorobenzene
48	3,3,5-trimethylheptane	805	Phenanthrene	1574	P-dichlorobenzene
49	2,4,4-trimethylhexane	806	Chrysene	1592	1,2,4-trichlorobenzene
50	3,3-diethylpentane	807	Pyrene	1680	Bromobenzene
51	2,2,3,3-tetramethylpentane	808	Acenaphthene	1681	Methyl iodide
52	2,2,3,4-tetramethylpentane	820	Indane	1691	Iodobenzene
53	2,2,4,4-tetramethylpentane	899	Nitrous oxide	1701	Methylamine
54	2,3,3,4-tetramethylpentane	900	Nitrogen dioxide	1703	Trimethylamine
55	Squalane	901	Oxygen	1704	Ethylamine
56	N-decane	902	Hydrogen	1706	Triethylamine
62	Tert-butylcyclohexane	904	Nitrogen trioxide	1710	Diethylamine
63	N-undecane	905	Nitrogen	1722	Methyl DiEthanolAmine
64	N-dodecane	906	Nitrogen tetroxide	1723	Monoethanolamine
65	N-tridecane	908	Carbon monoxide	1724	Diethanolamine
66	N-tetradecane	909	Carbon dioxide	1725	Triethanolamine
67	N-pentadecane	910	Sulfur dioxide	1741	Ethylenediamine
68	N-hexadecane	911	Sulfur trioxide	1743	Diisopropylamine
69	N-heptadecane	912	Nitric oxide	1750	N-aminoethyl piperazine
70	N-octadecane	913	Helium-4	1760	Nitromethane
71	N-nonadecane	914	Argon	1761	Nitroethane
72	2,2-dimethyloctane	915	Air	1762	1-nitropropane
73	N-eicosane	917	Fluorine	1763	2-nitropropane
74	N-heneicosane	918	Chlorine	1769	1-nitrobutane
75	N-docosane	919	Neon	1771	Hydrogen cyanide
76	N-tricosane	920	Krypton	1772	Acetonitrile
77	N-tetracosane	922	Bromine	1773	Propionitrile
78	N-pentacosane	924	Ozone	1774	Acrylonitrile
79	N-hexacosane	959	Xenon	1775	Methacrylonitrile
80	N-heptacosane	1001	Formaldehyde	1778	O-nitrotoluene
81	N-octacosane	1002	Acetaldehyde	1779	P-nitrotoluene
82	N-nonacosane	1003	Propanal	1780	M-nitrotoluene
85	3-methylnonane	1005	Butanal	1791	Pyridine
86	2-methylnonane	1006	2-methylpropanal	1792	Aniline
87	4-methylnonane	1007	Pentanal	1801	Methyl mercaptan
88	5-methylnonane	1008	Heptanal	1802	Ethyl mercaptan
91	2-methyloctane	1009	Hexanal	1803	N-propyl mercaptan
92	3-methyloctane	1051	Acetone	1804	Tert-butyl mercaptan
93	4-methyloctane	1052	Methyl ethyl ketone	1805	Isobutyl mercaptan
94	3-ethylheptane	1053	3-pentanone	1806	Sec-butyl mercaptan
96	2,2-dimethylheptane	1054	Methyl isobutyl ketone	1807	N-hexyl mercaptan
102	Cyclobutane	1057	3-heptanone	1810	Isopropyl mercaptan
104	Cyclopentane	1058	4-heptanone	1813	Methyl ethyl sulfide
105	Methylcyclopentane	1059	3-hexanone	1814	Methyl n-propyl sulfide
107	Ethylcyclopentane	1060	2-pentanone	1815	Methyl t-butyl sulfide

108	1,1-dimethylcyclopentane	1061	Methyl isopropyl ketone	1816	Methyl t-pentyl sulfide
109	Cis-1,2-dimethylcyclopentane	1062	2-hexanone	1817	Di-n-propyl sulfide
110	Trans-1,2-dimethylcyclopentane	1063	2-heptanone	1818	Diethyl sulfide
111	Cis-1,3-dimethylcyclopentane	1064	5-methyl-2-hexanone	1820	Dimethyl sulfide
112	Trans-1,3-dimethylcyclopentane	1066	3,3-dimethyl-2-butanone	1821	Thiophene
114	N-propylcyclopentane	1068	Diisobutyl ketone	1824	Diethyl disulfide
115	Isopropylcyclopentane	1069	Diisopropyl ketone	1828	Dimethyl disulfide
116	1-methyl-1-ethylcyclopentane	1080	Cyclohexanone	1829	Di-n-propyl disulfide
122	N-butylcyclopentane	1100	Ketene	1844	Dimethyl sulfoxide
137	Cyclohexane	1101	Methanol	1845	Sulfolane
138	Methylcyclohexane	1102	Ethanol	1851	Acetyl chloride
140	Ethylcyclohexane	1103	1-propanol	1854	Dichloroacetyl chloride
141	1,1-dimethylcyclohexane	1104	Isopropanol	1855	Trichloroacetyl chloride
142	Cis-1,2-dimethylcyclohexane	1105	1-butanol	1876	N,n-dimethylformamide
143	Trans-1,2-dimethylcyclohexane	1106	2-methyl-1-propanol	1886	Nitrobenzene
144	Cis-1,3-dimethylcyclohexane	1107	2-butanol	1889	Furfural
145	Trans-1,3-dimethylcyclohexane	1108	2-methyl-2-propanol	1893	Carbonyl sulfide
146	Cis-1,4-dimethylcyclohexane	1109	1-pentanol	1894	Phosgene
147	Trans-1,4-dimethylcyclohexane	1110	2-pentanol	1903	Nitric acid
149	N-propylcyclohexane	1111	2-methyl-2-butanol	1904	Hydrogen chloride
152	N-butylcyclohexane	1112	2-methyl-1-butanol	1907	Hydrogen iodide
153	Cis-decahydronaphthalene	1113	2,2-dimethyl-1-propanol	1911	Ammonia
154	Trans-decahydronaphthalene	1114	1-hexanol	1921	Water
201	Ethylene	1125	1-heptanol	1922	Hydrogen sulfide
202	Propylene	1151	Cyclohexanol	1938	Carbon disulfide
204	1-butene	1181	Phenol	1940	Sulfur hexafluoride
205	Cis-2-butene	1182	O-cresol	2252	2-methyl-1-heptene
206	Trans-2-butene	1183	M-cresol	2367	Propylene carbonate
207	Isobutene	1184	P-cresol	2391	Dimethyl carbonate
209	1-pentene	1201	Ethylene glycol	2717	Diethylenetriamine
210	Cis-2-pentene	1202	Diethylene glycol	2732	N-aminoethyl ethanolamine
211	Trans-2-pentene	1203	Triethylene glycol	2743	2,4-dinitrotoluene
212	2-methyl-1-butene	1204	Tetraethylene glycol	2744	2,6-dinitrotoluene
213	3-methyl-1-butene	1231	Glycerol	2745	3,4-dinitrotoluene
214	2-methyl-2-butene	1241	1,4-butanediol	2747	2,4,6-trinitrotoluene
216	1-hexene	1252	Acetic acid	2748	2,5-dinitrotoluene
217	Cis-2-hexene	1253	Propionic acid	2749	3,5-dinitrotoluene
218	Trans-2-hexene	1255	Oxalic acid	2750	P-phenylenediamine
221	2-methyl-1-pentene	1256	N-butyric acid	2752	Piperazine
227	4-methyl-cis-2-pentene	1277	Acrylic acid	2856	N,n-dimethylacetamide
228	4-methyl-trans-2-pentene	1278	Methacrylic acid	3801	Di-tert-butyl disulfide
234	1-heptene	1281	Benzoic acid	3813	Ethyl methyl disulfide
250	1-octene	1282	O-toluic acid	3814	Ethyl propyl disulfide
259	1-nonene	1283	P-toluic acid	3819	Diphenyl disulfide

261	1-undecene	1284	Salicylic acid	4865	Trichloroacetaldehyde
270	Cyclohexene	1285	Adipic acid	4868	Dichloroacetaldehyde
301	Propadiene	1286	Maleic acid	6861	Diethylethanolamine
302	1,2-butadiene	1287	Phthalic acid	6862	Methylethanolamine
303	1,3-butadiene	1289	Terephthalic acid	6863	Dimethylethanolamine
309	Isoprene	1291	Acetic anhydride	6864	Diisopropanolamine
316	Dicyclopentadiene	1298	Maleic anhydride	13125	DiPhenyl Carbonate
401	Acetylene	1301	Methyl formate	20101	2-Methyl-2-Heptanol
402	Methylacetylene	1302	Ethyl formate	22158	2-Methoxy-2-Methyl-Heptane
403	Ethylacetylene	1303	N-propyl formate	22587	Ethyl Phenyl Carbonate
404	Dimethylacetylene	1312	Methyl acetate	23498	Methyl Ethyl Carbonate
418	Vinylacetylene	1313	Ethyl acetate	27991	Methyl Phenyl Carbonate
501	Benzene	1314	N-propyl acetate	28366	DiEthyl Carbonate
502	Toluene	1315	N-butyl acetate		
504	Ethylbenzene	1316	Isobutyl acetate		